



ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: ogu.ahmet.topcu@gmail.com

Web: <http://mmf2.ogu.edu.tr/atopcu>

Bilgisayar Destekli

Nümerik Analiz

Ders notları 2014

Ahmet TOPÇU

```

C:\Basic\QBasic.EXE
File Edit View Search Run Debug Calls Utility Options Help
POWERM1.BAS:PowerMises
DEFINT I-N
DEFDBL A-H, O-Z
SUB PowerMises (n, a(), aLandaMax, x(), iIterate, iHata)
' Mutlak değerce En büyük özdeğer ve özvektör hesabı (PowerMises)
'
' Dr. ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' A(n,n) matrisinin mutlak değerce en büyük özdeğeri ve bu
' özdeğere ait x(n) normalleştirilmiş özvektörü hesaplanır.
' A simetrik olabilir veya simetrik olmayabilir.
' aLandaMax : A matrisinin mutlak değerce en büyük özdeğeri
' x(n) : Lamda ya ait normalleştirilmiş özvektör
' A matrisi çağırılan programda depolanmış ve x vektörü
' boyutlandırılmış olmalıdır.
' Maxit : maksimum iterasyon sayısı, aşılırsa iHata=1 döner
' Eps : hesap hassasiyeti
'
DIM y(n)
Maxit = 200: ' Maximum iterasyon sayısı
Eps = 1E-14: ' Hassasiyet
iHata = 0
FOR i = 1 TO n
  x(i) = 1
NEXT i
aLandaMax = 0: iIterate = 0: iState = 1
WHILE iIterate <= Maxit AND iState = 1
  AnewLanda = 0
  FOR i = 1 TO n
    y(i) = 0
    FOR j = 1 TO n
      y(i) = y(i) + a(i, j) * x(j)
    NEXT j
    IF ABS(y(i)) > ABS(AnewLanda) THEN AnewLanda = y(i)
  NEXT i
  Epsreached = ABS(aLandaMax - AnewLanda)
  aLandaMax = AnewLanda
  aNorm = 0
  FOR i = 1 TO n
    T = y(i) / AnewLanda
    aNorm = aNorm + (x(i) - T) ^ 2
  NEXT i
  x(i) = T
  NEXT i
  A x = λ x

```

28

PROGRAMLAR: Standart Özdeğer ve özvektör hesabı

- En büyük özdeğer ve vektörü - PowerMises

28. PROGRAMLAR: Standart özdeğer ve özvektör hesabı¹ - PowerMises

PowerMises alt programı

$$\underline{A}x = \lambda x$$

Standart özdeğer probleminin **mutlak değerce en büyük özdeğerini** ve buna ait normalleştirilmiş özvektörü Power (diğer adı: Mises² veya vektör) iterasyon yöntemi ile hesaplar. $\underline{A}_{n \times n}$ simetrik olmayabilir. Mutlak değerce en büyük özdeğere **dominant özdeğer** de denir.

PowerMises alt programı $\underline{A}(n,n)$ matrisinin çağırın programda depolanmış ve $\underline{x}(n)$ vektörünün tanımlanmış olduğunu varsayar.

Örnek:

$$\underline{A} = \begin{bmatrix} 10 & 2 & 6 & 1 & 0 \\ 2 & 0 & 4 & 1 & 2 \\ 7 & -4 & 10 & 5 & 1 \\ -3 & 1 & -5 & 5 & 3 \\ 4 & 3 & 1 & 3 & 5 \end{bmatrix}, \text{ en büyük özdeğer ve buna ait özvektör ?}$$

PowerMises alt programının sonucu

```

C:\Basic\QBASIC.EXE
İterasyon sayısı: 48
Özdeğer ve özvektörü(PowerMises):
Mutlak değerce en büyük özdeğer= 14.6231160474499
özvektör:
1 .363322587049547 .735722492927579 -.517864084214652 .443941232794793
  
```

Mutlak değerce en büyük özdeğer: $\lambda = 14.623116$. λ ya ait özvektör: $\underline{x} = \begin{bmatrix} 1 \\ 0.3633 \\ 0.7357 \\ -0.5179 \\ 0.4439 \end{bmatrix}$

\underline{A} matrisi tekrarlanan özdeğerlere, ($|\lambda_i| = |\lambda_j|$) sahipse veya bazı özdeğerler birbirine çok yakın ise ($|\lambda_i| \approx |\lambda_j|$) yakınsamayabilir. Bu gibi durumlarda özdeğer bulunsa bile özvektör hatalı olabilir. Tüm özdeğerlerin sanal olması durumunda ise Power metodu kesin yakınsamaz. Buna karşın, \underline{A} simetrik ve pozitif tanımlı ise daima yakınsar.

Örnek:

$$\underline{A} = \begin{bmatrix} 1 & 2 \\ -15 & 10 \end{bmatrix} \quad \underline{A} \text{ Matrisinin her iki özdeğeri de sanaldır: } \lambda_1 = 5.5 + 3.1225i \text{ ve } \lambda_2 = 5.5 - 3.1225i.$$

PowerMises yakınsamaz.

```

Microsoft QuickBASIC
İterasyon sayısı: 201
Max iterasyon sayısı aşıldı!
  
```

PowerMises alt programının sonucu

Bazı mühendislik problemlerinde mutlak değerce en büyük özdeğer yerine mutlak değerce en küçük özdeğerin ve buna ait normalleştirilmiş özvektörün hesaplanması gerekir (Örnek: bir yapının birinci periyodunun ve modunun hesabı). Bu durumda PowerMises programı yerine bölüm 29 da verilen PowerTers alt programı kullanılabilir.

¹ Teori ve örnekler için bak: bölüm 25

² Mises iterasyon yöntemi özdeğer hesaplayan ilk nümerik algoritmadır. 1929 yılında Avusturyalı Richard Edler von **Mises** tarafından geliştirildi

```

' -----Ana program PowerMises-----
' Mutlak değerce en büyük özdeğer ve buna ait özvektörün hesabı
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' -----

DATA 5: ' A nın boyutu
' A matrisi:
DATA 10, 2, 6, 1, 0
DATA 2, 0, 4, 1, 2
DATA 7, -4, 10, 5, 1
DATA -3, 1, -5, 5, 3
DATA 4, 3, 1, 3, 5

DATA 2
DATA 2,4
DATA 4,8

DEFINT I-N
DEFDBL A-H, O-Z
DECLARE SUB PowerMises (n, a(), aLamdaMax, x(), iIterate, iHata)
CLS

READ n
DIM a(n, n), x(n)
'A yı oku:
FOR i = 1 TO n
FOR j = 1 TO n
READ a(i, j)
NEXT j
NEXT i

CALL PowerMises(n, a(), aLamdaMax, x(), iIterate, iHata)

PRINT "iterasyon sayısı: "; iIterate
IF iHata <> 0 THEN PRINT "Max iterasyon sayısı aşıldı(PowerMises) ": END

PRINT "Özdeğer ve özvektörü(PowerMises):"
PRINT "Mutlak değerce en büyük Özdeğer="; aLamdaMax
PRINT "Özvektör:"
FOR i = 1 TO n
PRINT x(i);
NEXT i
PRINT

END ' PowerMises ana

```

PowerMises ana programı

A'nın satırları

A'nın boyutu okunuyor

A matrisi ve x özvektörü için bellekte yer ayrılıyor

A okunuyor

PowerMises alt programı çağırılıyor

Sonuçlar yazdırılıyor

```

SUB PowerMises (n, a(), aLamdaMax, x(), iIterate, iHata)
' Mutlak değerce En büyük özdeğer ve özvektör hesabı (PowerMises)
' -----
' Dr. ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' A(n,n) matrisinin mutlak değerce en büyük özdeğeri ve bu
' özdeğere ait x(n) normalleştirilmiş özvektörü hesaplanır.
' A simetrik olabilir veya simetrik olmayabilir.
' aLamdaMax : A matrisinin mutlak değerce en büyük özdeğeri
' x(n) : Lamda ya ait normalleştirilmiş özvektör
' A matrisi çağırılan programda depolanmış ve x vektörü
' boyutlandırılmış olmalıdır.
' Maxit : maksimum iterasyon sayısı, aşılsa iHata=1 döner
' Eps : hesap hassasiyeti
' -----

DIM y(n)
Maxit = 200: ' Maximum iterasyon sayısı
Eps = 1E-14: ' Hassasiyet
iHata = 0

FOR i = 1 TO n
x(i) = 1
NEXT i
aLamdaMax = 0: iIterate = 0: iState = 1
WHILE iIterate <= Maxit AND iState = 1
AnewLamda = 0
FOR i = 1 TO n
y(i) = 0
FOR j = 1 TO n
y(i) = y(i) + a(i, j) * x(j)
NEXT j
IF ABS(y(i)) > ABS(AnewLamda) THEN AnewLamda = y(i)
NEXT i

Epsreached = ABS(aLamdaMax - AnewLamda)
aLamdaMax = AnewLamda

```

PowerMises alt programı

```

aNorm = 0
FOR i = 1 TO n
T = y(i) / AnewLamda
aNorm = aNorm + (x(i) - T) ^ 2
x(i) = T
NEXT i
' istenilen hassasiyete ulaşıldı mı?
aNorm = SQR(aNorm)
IF Epsreached < aNorm THEN Epsreached = aNorm
iState = 0
IF Epsreached > Eps THEN iState = 1
iIterate = iIterate + 1
WEND

' Maxit aşıldı mı?
IF iIterate > Maxit THEN iHata = 1

END SUB 'PowerMises

```